

サイエンスツールとしてのパソコンの活用法に関する研究

—MSX パソコンによる簡易型自動気象観測装置 (MAMEDAS) —

理科教育研究室 杉 本 良 一

An Application of Personal Computer as a Science Tool

—Micro-automated Meteorological Data Acquisition System(MAMEDAS)

by an MSX computer and a simple interface—

Ryoichi SUGIMOTO

1 はじめに

今回改訂された学習指導要領の中学校及び高等学校理科においては、観察・実験の一層の重視が強調され、また、情報化への対応として、コンピュータを観察・実験の道具として使用するよう改訂がなされている¹⁾²⁾。中学校の学習指導要領理科の「第3 指導計画の作成と内容の取り扱い」では、「各分野の指導に当たっては、観察、実験の過程での情報の検索、実験データの処理、実験の計測などにおいて、必要に応じ、コンピュータ等を効果的に活用するよう配慮するものとする」とあり、また、高等学校理科では、物理 I B、化学 I B、生物 I Bの「探究活動」などにおいて、適宜コンピュータの活用を図ること。また、IIの科目では「課題研究」において、科学的に探究する方法を習得させ、問題解決の能力を育成するとともに、「解決すべき課題についての情報の検索、計測、結果の集計・処理などに、適宜コンピュータなどを活用させること」とある。パソコンを観察・実験の道具すなわちサイエンスツールとして用いるという内容は今回の改訂の大きな特徴である³⁾⁴⁾。

理科にパソコンを利用することは今後とも不可欠な事であると予想されるが、どのような使い方が最も適切であるかはあまり明らかにされていない。理科教育は、観察・実験などの直接経験を重視し、自分で考え、判断して、問題を解決する能力を高める必要があると考えるが、現在のコンピュータの理科への利用の仕方は、知識定着型のものやシミュレーションなどが多くみられ、探究の技能を身に付けさせるために、子ども自らが問題を発見したり、解決したりするような利用の仕方は少ないように思われる。

実験計画を立て、情報を収集したり、発見や判断をしたり、また、実験や測定をしてデータの解釈をするなど、子ども自らが問題解決をするようなパソコンの使い方がより望ましいと考える。実験室で実験可能な内容までも、チュートリアル型 CAI などにより、パソコンのディスプレイ上だけで理科授業をすることはあまりよい方法ではないと考える。

筆者は、データ検索、実験できない現象のシミュレーションやデータのグラフ化、表計算、ワープロソフトによるレポート作成などもパソコンの一つの利用の仕方であるが、観察・実験などを重視する理科の授業においては、時間や温度、電圧などを直接測定する計測機器として使い、また、

データの表示，計算，グラフ化などの処理及びデータの保存などに用いることがより適切な使い方ではないかと考える。

今回はその一例として，安価な MSX パソコンを用い，簡単なインターフェースによる簡易型の自動気象観測装置を開発し，中学校及び高等学校の気象学習に活用がはかれるよう工夫したので報告する。

2 気象教材の問題点とパソコン活用の意義

現行の学習指導要領中学校第2分野の気象現象の取り扱いでは，

(4)天気の変化—観測や実験を通して，天気の変化は太陽放射に基づく水の状態変化や大気の動きに関連して起こることを考察させ，それらをもとにして天気変化の仕組みや規則性を理解させる。

とあり，指導書⁵⁾では，「導入に当たっては，毎日の天気変化や四季に特有な気象を取り上げて天気の様子に関心をもたせ，授業の展開には，1～2週間程度連続して収集した気温，湿度，気圧，風などの観測データや，新聞天気図などを用意し，天気についての現象を身近なものとして動的にとらえさせることが必要である。」とあり，実際の観測データを用いることを強調している。

新指導要領では

(4)天気とその変化—身近な気象の観察，観測を通して，天気変化の規則性に気付かせるとともに，様々な気象情報を活用した天気の予測の方法について理解させ，天気変化についての認識を深める。

とある。さらに，小項目ア 天気の変化では「(7)校庭などで気象観測を行い，その記録に基づいて，天気変化の規則性を見いだすこと。」とある。

新指導要領では，身近な気象観測や，衛星画像の気象情報など，より一層の具体的データに基づく気象学習が強調されている。しかし，実際は学校で継続的に気象データをとったり，気象観測の施設を維持・管理したりすることは，大変な困難を伴うことが多い。また，現在の学校にある備品としての気象観測の機械・器具は，過去の理科教育振興法によって購入され，古くなったもの，故障したものが多いのが実情である。新規にこれらの備品を購入しようと思っても，乏しい学校の予算では，購入もままならないのが現状であろうかと察せられる（例えば自記温度計を例にとると，約5万円程度もかかる）。さらに，市販のコンピュータ制御の自動気象観測装置は数百万円もかかり，学校で購入することは不可能であろう。

「改訂中学校学習指導要領の展開」で小林は⁶⁾，上の小項目(7)の内容について，「できれば，年間を通じて気象要素の資料が欲しい。そのためには，自記の観測器が必要となる。かつてのゼンマイ式のものには保守管理が大変であった。もっと容易に利用できる自記の観測機械（安価で丈夫な）の出現が切に望まれるところである。」と書いている。

近年，比較的安価なパソコンなどが容易に入手できるようになった。これを用いて，教材用の簡単な気象観測装置を製作し，上記のねらいに沿った自記の観測システムを製作し，理科の授業に活

用することは大変意義深いと考える。

しかし、パソコンを用いて気象データを取り入れるためのハードウェアやソフトウェアを教師が自作するには、次のような問題点があると考えられる。

- ①パソコンについての高度なハードウェアの知識や、マシン語などに習熟している必要がある。
- ②市販のA/Dコンバータなどのインターフェースは高価であり、自作することも容易ではない。
- ③あまり複雑なものでは生徒が工夫したり、改良したりする余地がない。

そこで、これらの問題点を取り除いた気象観測データが自動的に記録できるシステム(マメダス, MAMEDAS, Micro Automated Meteorological Data Aquisition System)を開発した。これを用いれば、生徒の意欲的な学習が期待できるとともに、生徒自身にハードウェアの改良やプログラムの作成などをさせれば、コンピュータリテラシーを養うことができ、さらに、実験計画や測定方法の工夫、情報の収集、データの解釈など、科学的方法を学ばせるのにもよい教材になると考える。

3 システムの概要

本教材は生徒が工夫をしながら気象観測システムを製作していく過程で、様々な科学的手法を学習できることをねらっている。また、どんな機種のパソコンでも応用できる技能や知識が身に付けられると考える。さらに、較正をすることにより、精度をあげれば実用にも役立つものである。

測定の内容は

① 気温 (0-40°C) ② 湿度 (0-100%) ③ 風速 (相対量) ④ 日照 (相対量) の4項目である。気象要素のうち、気圧の測定には増幅器や8ビットのA/Dコンバータなどが必要となり、高価となること及び生徒が扱うには複雑すぎるので今回は省略した。

センサーとしては、気温、湿度にサーミスタ、風速計にはフォトランジスタ、日照にCdSを用いた。データはMSX-BASICのPDL及びSTICK関数を使い⁷⁾⁸⁾⁹⁾、0-255の整数値でデータを取り出す。

パソコンとしてはMSX₂規格で、3.5インチ2DDフロッピーディスクドライブを一台内蔵したHB




図1 マメダスシステムの概要

-F500 (SONY 製) を用いた。この規格のパソコンではリアルタイムクロックが付いている。そのため、日付と時刻が測定でき、データの記録に便利が良い。

図1にこのシステム構成の概要を示す。

図2, 3にマメダスシステムの全体写真とセンサー部分の写真を示す。




図2 マメダスシステムの全景




図3 センサー部分の拡大写真

4 測定原理と較正方法

(1) 気温の測定

気温測定用のインターフェースについては、PDL関数を用いるが、その入力回路にはタイマー IC555を用いる。図4の回路で、コンデンサーの値を一定にしておき、抵抗の代わりにサーミスタを接続する。サーミスタは温度によって、その抵抗値が変化するので、これを用いて温度を測定することが可能となる。

いま、 T_0 [K] におけるサーミスタの抵抗値を R_0 [Ω]、 T [K] における抵抗値を R [Ω] とすると、

$$R = R_0 \exp \{ B (1/T - 1/T_0) \} \quad \dots\dots\dots ①$$

ここで、 B はサーミスタによる定数で、狭い温度範囲ならば一定である。

タイマー IC の出力パルス幅 t [sec] は、コンデンサーの容量を C [F] とすると、

$$t = 1.1 \times R \times C \quad \dots\dots\dots ②$$

また、MSX-BASIC の PDL 関数の値を T [K] のとき n 、 T_0 [K] のとき n_0 とすると、 n は t や R に比例する。すなわち

$$n \propto t \propto R \quad \dots\dots\dots ③$$

したがって、測定温度 T [K] は、(1)~(3)より





図4 インターフェース回路

$$T = B / \{ \ln(n/n_0) + B/T_0 \} \quad \dots\dots\dots ④$$

B, n_0 , T_0 は定数とみなせるので, PDL 関数 n の値により温度 T を求めることができる。

図5は, サーミスタ SDT1000 (東京三洋) を用い, 1リッターピーカーにおいて, 温度を低温側から, 約5°Cずつ上昇し, 標準温度計と比較したものである。このとき B 定数=4265, $n_0=142$ であった。高温や低温側で最大で約1.3°Cの誤差があるが, これは, B 定数が温度によって変わるためであり, 厳密には温度により B 定数を変えなければならない。しかし, プログラムや較正が複雑となるので, このシステムでは温度の測定範囲内で一定とみなした。教材用としては使う限りではあまり支障ないと考える。あくまで, 安価にできることをねらっており, 精度をあげようとする, 高価なサーミスタやインターフェースが必要となる。



△サーミスタの温度 --- 標準温度計の読み

図5 マメダスシステム温度計の誤差

サーミスタや CdS の抵抗は数 K Ω から数十 K Ω

あり, センサーは数百メートル離れていても測定可能である。すなわち, リモートセンシングができるのもこのシステムの特徴である。また, 5ボルトのパルス駆動なので, 他のセンサーに比べ, 雑音の影響も受けにくい。

(2)湿度の測定

湿度については, 理科年表に掲載されている公式を参考に, 近似式を作り, また, 乾球温度と湿球温度を測ることにより測定ができる。センサーはサーミスタをエポキシ系接着剤で固め, 湿らせたガーゼを取り付け, 湿球温度を計る。

すなわち, 相対湿度 H (%) は

$$H = e / E_w \times 100 \quad \dots\dots\dots ⑤$$

ここで E_w は $t^\circ\text{C}$ における飽和水蒸気圧, e は $t^\circ\text{C}$ における水蒸気圧であり,

$$e = E_w' - 0.5 (t - t') P / 755 \quad \dots\dots\dots ⑥$$

ただし, $\left[\begin{array}{l} t : \text{乾球の示度} \quad t' : \text{湿球の示度} \\ E_w' : t^\circ\text{C} \text{ における飽和水蒸気圧} \quad P : \text{大気圧} \end{array} \right]$

E_w の値を気温の関数として近似的に次のような多項式で表した。

$$E_w = 6.1 + 0.37 t + .022 t^2 + .000227 t^3 \quad \dots\dots\dots ⑦$$

図6に, 水蒸気の飽和蒸気圧曲線と⑦式による曲線を示す。高温側で誤差が大きくなるが, その差は4%程度である。

湿度の較正は, 60%付近では室温中で, また, 30%付近では密閉したスチロール容器中に, 飽和塩化マグネシウム溶液400cc満たし, さらに, 90%付近では, 湯をいれ, 市販の温湿度計 HT150 (ソアー) との比較によった¹⁰⁾。HT150は3500円程度で入手でき, その精度は $\pm 5\%$ (30-80%) である。図7にマメダスと比較したものを示す。また, 風が強くと吹く場合大きく湿度が低下する。これはサーミスタが敏感に反応するためである。このマメダスの精度は, $\pm 10\%$ 程度と考えられるが, 精密な測定はきわめて難しく, 相対的に湿度が高いか低いかが分かると考えた方がよい。




図6 飽和水蒸気圧曲線




図7 湿度の較正 (HT150との比較)

(3)風速の測定

ゲームなどに用いるジョイスティック端子は、各ピンの電圧の状態により、外部からの信号を取り入れることができる。ここでは、図8のような使い捨て注射器やフィルムケースなどを使った簡単な風速測定器を製作し、その回転数をマイクロスイッチ、リードスイッチまたはフォトランジスタなどのセンサーでカウントする。風速は、別の携帯用風速計などで較正する。




図8 風速計センサーの構造




図9 風速計センサーの写真

(4)日照の測定

日照の測定にはCdSセルを用いた。このインターフェースは温度測定と同じ回路(図4)を用いる。サーミスタの代わりに接続する。光の強弱によって、CdSの抵抗値が変化することにより、PDL関数の数値が変化することを利用する。但し、CdSの抵抗値は、サーミスタと異なり、大きく変化するので、コンデンサーや直列に変抵抗器をいれて調節する。なお、太陽光を直接当てると強すぎるので、フェルトペンなどで黒く塗る。較正は照度計などを用いるとよいが、広い範囲の光の強さには対応しきれないので、くもりの日に中間の値を示すような場所に設置する。

5 ソフトウェア

メニュー画面は図10のようになり、始めに4 初期設定で記録するデータのファイルネーム FI\$, 測定時間の間隔 DT (sec), 風速測定の時間 T (sec) を入力しておく。次に1の測定を選択すると測定が開始される。

記録データ形式はランダムファイルで表1のように1つのデータは32バイトで記録される。したがって、720K バイトのフロッピーには、約2万データとることができる。

表1 マメダスデータの形式

日付	時刻	気温	湿度	風速	日照
DA\$(8バイト)	TI\$(8バイト)	TI(4バイト)	HU(4バイト)	WV(4バイト)	SL(4バイト)

停電などでパソコンの電源が切れても、データは消えることはない。データはグラフィック画面に表示することができ、気象データの時間変化が分かりやすく表示され、理解しやすい。

また、3を選ぶことによりプリンターにデータを打ち出すこともできる。プログラムはすべて BASIC 言語で書かれており、教師や生徒自身が改良して行くことにより様々な応用が考えられる。

このプログラムリストは、資料1に示す。

```

==MAMEDAS==
      MENU
1     ソクテイ
2     データヒョウジ
3     データプリント
4     ショキセツテイ
5     オワリ
  
```

図10 メニュー画面




図11 測定結果例

6 測定結果例と教材としての活用法

図11は、広島県立教育センター科学芸術教育棟物理研究準備室北側のベランダにおいて、1988年11月18日から約1週間に渡り、継続して測定した画面である。データは10分おきに、853データとられたものである。この時期では、昼間の日照時間が短いことが一目で分かる。また、夜には気温が下がると同時に湿度が上昇しているのがわかる。年間を通して記録すれば、さらに様々な情報が得られると考える。

教材に当たっては、以下に述べるような方法でこのシステムを活用することにより、生徒の興味・関心を高め、また、観察・実験の効率化を計ることができると考える。

高等学校地学の気象学習でも利用することができるが、ここでは中学校第2分野の気象学習における活用について例をあげる。

(1) 校内における気温、湿度、日照の日変化の測定

学校内で、日向や日陰、教室の中やグラウンドといったふうに、色々な場所でのマイクロ気象のデータを記録することにより、その環境との因果関係から気象学習に興味・関心を持たせることができる。

(2) 毎日の気温、湿度、日照、風力の記録

日々の気象データを自動記録しておき、一ヶ月毎にまとめる。そのとき同時に新聞などの天気予報欄を同時に切抜きしておく。そして、後に授業の教材として活用する。

(3) 1日の気温変化と日照の関係を調べる

太陽の日射は南中時が最大となるが、気温はそれよりも遅れることなどを確かめる。すなわち、気温の変化と日照センサーのデータを比べる。

(4) 風が強い時の湿度変化を調べる

風の強さを風速計センサーで計測し、そのときの湿度の変化を読み取る。また、室内などでエアコンを作動させたときの、室温と湿度の関係などを調べる。

(5) 1年間のデータを比べ、季節による違いを比較する

1年間の気象データを蓄積し、季節による気温や湿度の変化、風速や、日照などを比較し、季節の特徴をまとめる。

(6) 前線の移動の観測

水銀気圧計などの変化とこのマメダスシステムの気温・湿度変化の関係から、前線などの移動の様子を調べる。

(7) 気象衛星画像とマメダスデータの比較

気象衛星ひまわりの雲画像データとマメダスシステムによる気温、日照データを比較し、マイクロの気象とマクロの気象を関係付ける。

7 おわりに

このシステムは筆者がタイ国に理科教育の技術指導の過程で、タイ国教員養成大学の教官にパソコンの理科教育への活用を指導する目的で開発・製作したものである⁷⁾。このシステムを作り、指導する過程で、様々な科学的手法が学習できることに気付いた。指導した教官は、このシステムを組

み立てる過程を学習した後、アップルIIやIBMPC-XTなど他機種のパソコンでもこのシステムを移植することができた。

このシステムは研究用の高精度のものを考えてはいないので、耐久性や精度の面でまだ改良しなければならない部分もあるが、較正を正確にして、精度をあげれば実用的に役立つシステムにすることができると考える。また、このシステムは気象データがグラフィック画面に時間変化として表示され児童・生徒にとって理解しやすく、気象学習への興味・関心を高めることができる。

プログラムはすべてBASIC言語で書かれており、中学以上であれば生徒自身がプログラムを改良することもできるであろう。また、理科だけでなく、中学校の技術・家庭科「情報基礎」の領域と関連させて、生徒自らの手でマイコンを操作させたり、プログラムの作成、改良をさせたりでき、コンピュータリテラシーを養うことも可能である。さらに、マメダスシステムを生徒自身に製作させれば、実験計画、測定方法の工夫、情報やデータの収集など、科学の方法を学ばせるのに良い教材になると考える。

引用文献

- 1) 文部省 : 中学校学習指導要領, 1989, 大蔵省印刷局
- 2) 文部省 : 高等学校学習指導要領, 1989, 大蔵省印刷局
- 3) 山極 隆 : 教育と情報, No.359, P.20, 1988
- 4) 山極 隆 : NEW 教育とマイコン, P.34, P.58, 5月号, 1989
- 5) 文部省 : 中学校指導書 理科編, 1978 (昭和53年)
- 6) 小林 学 : 改訂中学校学習指導要領の展開-理科編- (小暮, 山極, 江田編), P.144, 明治図書, 1989
- 7) 杉本良一 : 「簡単なマイコンインターフェースによる物理実験法及び指導法の開発」
文部省科研費奨励研究B研究報告書, 1986 (昭和61年度)
- 8) 杉本良一 : 「簡単なマイコンインターフェースによる物理実験」, 物理教材の研究, 全国理科教育センター
研究協議会編, 東洋館出版社, 1988
- 9) 杉本良一 : 「マイコンを活用した熟教材の指導」, 理科の教育, P.42, 2月号, 1987
- 10) ソア-樹 : 温湿度計モジュール HT-150取り扱い説明書
- 11) 杉本良一 : 「タイの教育事情・教員養成大学の現状と技術協力」, 教育けんきゅう, 広島県立教育センター,
No.16, 1988,

(1990年8月31日受理)

資料1 マメダスシステムのプログラムリスト (MSX₂-BASIC)

```

100 ' SAYE "MAMEDAS"
110 ' MICRO AUTOMATED METEOROLOGICAL DATA ACQUISITION SYSTEM
120 ' PROGRAMED & EQUIPED BY Mr.Sugimoto, 1988.11.v2.0
130 CLS
140 MAXFILES=3 :RE=1
150 OPEN "GRP:" FOR OUTPUT AS #3
160 '----- デイズ -----
170 B1=4265:B2=4670:KT=273.15:LT=273.15+14.5:Q1=142:Q2=158:WV=0
180 '-----
190 KEY1,"ソクテイ":KEY2,"ヒョウシ":KEY3,"プリント":KEY4,"ショクセツテイ":KEY5,"オフリ"
200 KEY6,"MENU":KEY7,"MAMEDAS":KEY8,"SYSTEM":KEY9,"By.R.S"
210 ON KEY GOSUB 450,1030,1400,1600,1700,230
220 FOR K=1TO6:KEY(K)ON:NEXTK
230 REM MICRO AUTOMATED METEOROLOGICAL DATA ACQUISITION SYSTEM
240 FOR K=1TO6:KEY(K)ON:NEXTK
250 COLOR 15,4,7
260 SCREEN 0:WIDTH 40:CLS
270 LOCATE 0,0
280 PRINT" =====MAMEDAS===== "
290 PRINT"          メニュー
300 PRINT"
310 PRINT"          F1 ソクテイ
320 PRINT"          F2 データ ヒョウシ
330 PRINT"          F3 データ プリント
340 PRINT"          F4 ショクセツテイ
350 PRINT"          F5 オフリ F6 メニュー
360 PRINT" =====
370 PRINT:PRINT:PRINT
380 LOCATE 0,18
390 PRINT"MAMEDAS=Micro Automated Meteorological          Data Acquisition System"
400 PRINT
410 LOCATE 0,10:PRINT "F1-F6 ヲ センタク":PRINT
420 GOTO 420
430 ' *****
440 '
450 '          データ トリコミ
460 '
470 ' *****
480 SCREEN 0:WIDTH 40:CLS:TIME=0:RE=1
490 IF DT=0 OR TW=0 THEN 500 ELSE 510
500 PRINT "ショクセツテイイフシ ":FORI=1TO 600:NEXT I:GOTO 230
510 IF PIS="" THEN 520 ELSE 530
520 INPUT "ファイル名イフ イロ ":PIS
530 DEFN(T)=6.1+.32*T+.022*T*T+2.27E-04*T*T*T+1.2E-06*T^4
540 ' *****
550 SCREEN 2:CLS
560 OPEN "GRP:" FOR OUTPUT AS #3
570 FOR N=1TO 24:LINE(N*10,177)-(N*10,180),15:NEXT N
580 FOR N=1TO 24:LINE(N*10,50)-(N*10,47),15:NEXT N
590 FOR N=1TO 24:LINE(N*10,110)-(N*10,107),15:NEXT N
600 FOR N=1TO 24:LINE(N*10,150)-(N*10,147),15:NEXT N

610 FOR M=1TO 18:LINE(10,M*10)-(13,M*10),15:NEXTM
620 LINE(10,10)-(245,50),15,B
630 LINE(10,60)-(245,110),15,B
640 LINE(10,120)-(245,150),15,B
650 LINE(10,150)-(245,180),15,B
660 PSET(55,52),11:PRINT#3,"Micro AMEDAS SYSTEM"
670 PSET(0,10):PRINT#3,"*":PSET(0,25):PRINT #3,"+":PSET(0,40):PRINT#3,">"
680 PSET(0,65):PRINT#3,"<":PSET(0,76):PRINT #3,"^":PSET(0,90):PRINT#3,"~"
690 PSET(0,125):PRINT#3,"カ":PSET(0,135):PRINT #3,"ヒ"
700 PSET(0,150):PRINT#3,"ヒ":PSET(0,160):PRINT #3,"カ":PSET(0,170):PRINT#3,"リ"
710 OPEN "A:"*PIS* ".mda" AS #1 LEN=32
720 LINE(0,0)-(255,8),2,BF
730 LINE(20,182)-(169,191),4,BF
740 FIELD #1,8 AS DD$,8 AS TT$,4 AS T$,4 AS H$,4 AS W$,4 AS S$
750 ' -----
760 WV=0
770 GET DATE DA$:GET TIME TIS
780 P1=PDL(1)
790 P2=PDL(6)
800 P3=PDL(3)
810 T1=B1/(LOG(P1/Q1)*B1/LT)-KT
820 T2=B2/(LOG(P2/Q2)*B2/LT)-KT
830 E1=FNA(T1):E2=FNA(T2):HU=(E2-2.206*(T1-T2))/E1*100
840 SL=P3
850 ON INTERVAL=TW*60 GOSUB900
860 INTERVAL ON
870 IF STICK(1)=0 THEN GOTO 890 ELSE          WV=WV+1
880 IF STICK(1)=7 THEN GOTO 880 ELSE 870
890 WV=WV:GOTO 870
900 PSET(10+RE,50-T1),3:PSET(10+RE,110-HU/2),6:PSET(10+RE,150-WV),7:PSET(10+RE,150+SL/8.6),9
910 WV=WV-1
920 PSET(20,0):PRINT #3,USING" T1 ##.#C HU###.## WV### SL###":T1;HU;WV;SL
930 INTERVAL OFF
940 RSET DD$=DA$:LSET TT$=TIS
950 LSET T$=MKS$(T1):LSET H$=MKS$(HU)
960 LSET S$=MKS$(SL):LSET W$=MKS$(WV)
970 PUT #1,RE

```

```

980 CLOSE #1
990 RE=RE+1
1000 PSET(20,185):PRINT #3,DA$;" ";T1$
1010 PSET(170,185):PRINT #3,USING "time ####s";DT*TW
1020 IF TIME/6 MOD DT*10 THEN 1020 ELSE 710
1030 ' *****
1040 '
1050 '   テ-タ ヒヨウシ- グ-ラフ
1060 '
1070 ' *****
1080 IF F1$="" THEN 1090 ELSE 1100
1090 INPUT "ファイルメイ イレヨ ";F1$
1100 OPEN "a:"+F1$+".mda" AS #2 LEN=32
1110 FIELD #2,8 AS DD$,8 AS TT$,4 AS T$,4 AS H$,4 AS W$,4 AS S$
1120 INPUT "レコ-ト-スウ ステツプ-スウ イレヨ";RE,ST
1130 SCREEN 2
1140 FOR N=1TO 24:LINE(N*10,177)-(N*10,180),15:NEXT N
1150 FOR N=1TO 24:LINE(N*10,50)-(N*10,47),15:NEXT N
1160 FOR N=1TO 24:LINE(N*10,110)-(N*10,107),15:NEXT N
1170 FOR N=1TO 24:LINE(N*10,150)-(N*10,147),15:NEXT N
1180 FOR M=1TO 18:LINE(10,M*10)-(13,M*10),15:NEXT M
1190 LINE(10,10)-(245,50),15,B
1200 LINE(10,60)-(245,110),15,B
1210 LINE(10,120)-(245,150),15,B
1220 LINE(10,150)-(245,180),15,B
1230 PSET(55,52),11:PRINT#3,"Micro AMEDAS SYSTEM"
1240 PSET(0,10):PRINT#3,"キ":PSET(0,25):PRINT #3,"ヲ":PSET(0,40):PRINT#3,"ン"
1250 PSET(0,65):PRINT#3,"シ":PSET(0,75):PRINT #3,"ツ":PSET(0,90):PRINT#3,"ト"
1260 PSET(0,125):PRINT#3,"カ":PSET(0,135):PRINT #3,"ヒ"
1270 PSET(0,150):PRINT#3,"ヒ":PSET(0,160):PRINT #3,"カ":PSET(0,170):PRINT#3,"リ"
1280 PSET(20,185):PRINT #3,DD$;" ";TT$
1290 PSET(170,185):PRINT #3,USING "time ####s";DT
1300 FOR I=1 TO RE STEP ST
1310 GET #2,I
1320 II=INT(1/ST)+1
1330 PSET(11+10,50-CVS(T$))
1340 PSET(11+10,110-CVS(H$)/2)
1350 PSET(11+10,150-CVS(W$))
1360 PSET(11+10,150+CVS(S$)/8.5)
1370 NEXT I
1380 CLOSE #2
1390 AS=INKEY$:IF AS="" THEN 1390 ELSE 230
1400 ' *****
1410 '
1420 '   テ-タ フ-リント
1430 '
1440 ' *****
1450 SCREEN 0:WIDTH 40
1460 IF F1$="" THEN 1470 ELSE 1480
1470 INPUT "テ-タメイ イレヨ";F1$
1480 OPEN "a:"+F1$+".mda" AS #2 LEN=32
1490 FIELD #2,8 AS DD$,8 AS TT$,4 AS T$,4 AS H$,4 AS W$,4 AS S$
1500 INPUT "レコ-ト-スウ ステツプ-スウ イレヨ";RE,ST
1510 LOCATE 0,0:PRINT"DATE      TIME      TEMP      HUMID      W.V S.L"
1520 LPRINT"DATE      TIME      TEMP      HUMID      W.V S.L"
1530 FOR I=1 TO RE STEP ST
1540 GET #2,I
1550 PRINT USING " & & & ##.#C ##.#% ##.# ##.#";RIGHT$(DD$,5),LEFT$(TT$,5),CVS(T$),CVS(H$),CVS(W$),CVS(S$)
1560 LPRINT USING " & & & ##.#C ##.#% ##.# ##.#";RIGHT$(DD$,5),LEFT$(TT$,5),CVS(T$),CVS(H$),CVS(W$),CVS(S$)
1570 NEXT I
1580 CLOSE #2
1590 AS=INKEY$:IF AS="" THEN 1390 ELSE 230
1600 ' *****
1610 '
1620 '   ショキ セツテイ
1630 '
1640 ' *****
1650 CLOSE #1:CLOSE #2:CLOSE #3
1660 INPUT "   テ-タファイルメイ ";F1$
1670 INPUT "   シ-カンカンカク(sec) =";DT
1680 INPUT "   フウソク ソクテイノ シ-カン";TW
1690 GOTO 230
1700 REM ++++++ END ++++++
1710 REM *** リ-ミスタ ノ コウセイ ***
1720 SCREEN 0:WIDTH 40:CLS
1730 CLOSE #1:CLOSE #2:CLOSE #3
1740 FOR K=1 TO 6 :KEY(K)OFF:NEXT K
1750 KEY1," ":KEY2," ":KEY3," ":KEY4," ":KEY5,"run"+CHR$(13)
1760 END
1770 CLS
1780 B1=4265:B2=4570:KT=273.15:LT=273.15+14.5:Q1=142:Q2=158:WV=0
1790 DEFFNA(T)=6.1+.32*T+.022*T*T+2.27E-04*T*T*T+1.2E-06*T^4
1800 LOCATE 0,5
1810 P1=PDL(1)
1820 P2=PDL(5)
1830 P3=PDL(3)
1840 T1=B1/(LOG(P1/Q1)+B1/LT)-KT
1850 T2=B2/(LOG(P2/Q2)+B2/LT)-KT
1860 E1=FNA(T1):E2=FNA(T2):HU=(E2-2.206*(T1-T2))/E1*100
1870 H2=(E2-.8106*(T1-T2))/E1*100
1880 SL=P3
1890 PRINT P1,P2,P3
1900 PRINT USING "t1=##.# c t2=##.# c";T1;T2
1910 PRINT USING "シツト`=##.#% ニツショウ=##.# h2=##.#";HU;SL;H2
1920 GOTO 1800

```

SUMMARY

AN APPLICATION OF PERSONAL COMPUTER

AS A SCIENCE TOOL

-Micro-automated Meteorological Data Acquisition System(MAMEDAS)

by an MSX computer and a simple interface-

Ryoichi SUGIMOTO

Department of Science Education,

Faculty of Education,

Tottori University

Micro-automated Meteorological Data Acquisition System(MAMEDAS) was constructed. This is an example of application of personal computer in the field of scientific measurement for the secondary school science.

Firstly, one of the main difficulties in the field of meteorological teaching is to accumulate the meteorological data such as temperature, humidity, wind velocity and so on. If we use the micros, the difficulties are removed.

The second problem is that the teachers must learn about the interfacing technic of the sensors and the computer. However the MAMEDAS can be handled easily because of its simplicity.

The third problem is that the budget of schools is limited to purchase the commercial meteorological system. If a cheap MSX type computer is used, it is able to construct a simple and cost effective meteorological system.

The MAMEDAS system can measure,

- 1 Temperature (0-40 degrees centigrade)
- 2 Humidity (0-100 %)
- 3 Wind Velocity (relative value)
- 4 Sunlight intensity (relative value)

Temperature and Humidity are measured by thermister sensors. Sunlight intensity are measured by a CdS-photo cell. These data are aquisitted by PDL statement of MSX-BASIC. Wind Velocity is measured by a photo-transistor with a lamp and the data is aquisitted by STICK statement of MSX-BASIC.

As a consequence, the system is a typical model for learning the science method by using personal computer as a science tool. Data acquisition and interfacing technics are also applied to another fields of science education.